Cogging Test Switch Module

The Cogging Test Switch Module is used for testing an alternate Cogging controller during Booster operation.  The module supports switching between Primary and Alternate sets of 4 digital output signals and a Primary and Alternate analog signal (+/- 5 Volts).
When a Booster event trigger pulse is seen at the “Alternate I/O Select” input of the module, the 4 alternate digital signals and the alternate analog signal are switched onto the outputs of the module for 40 milli-seconds.  Otherwise, the primary 4 digital signals and the primary analog signal appear at the outputs.  Figure 1 indicates the connections on the front and rear of the module.
The module is a modified Dual Phase Detector Module Version 2 (phasedet2).   Figures 2 to 5 show the circuit modifications.  Beams-doc-3922-v2 can be referred to for the other details of the module.
The analog signals through the module can be calibrated with gain and offset adjustment set using the USB – PC interface (DLP).  A screenshot of this interface is shown in Figure 6.
The FPGA logic that controls the switching is given in Figure 7 and Listing 1.

Listing 1

Library IEEE;
USE IEEE.std_logic_1164.all;
USE IEEE.std_logic_arith.all;
USE ieee.std_logic_unsigned.all;

-- My Definitions --------------------------------------------------
USE work.pkg_081006.all;  -- my definitions and entities

entity Cogging_Test_Switch is
	port(clk        		: in std_logic;
		 nreset     		: in std_logic;
		 Enable_Test 		: in std_logic;
		 
		 DIN_1				: in std_logic;
		 DIN_2				: in std_logic;
		 DIN_3				: in std_logic;
		 DIN_4				: in std_logic;
		 
		 T1					: in std_logic;
		 T2					: in std_logic;
 		 T3					: in std_logic;
		 T4					: in std_logic;
		 
		 LED_PB_SW			: in std_logic;
		 SW_1					: in std_logic;
		 SW_2					: in std_logic;
		 
		 DO1					: out std_logic;
		 DO2					: out std_logic;
		 DO3					: out std_logic;
		 DO4					: out std_logic;
		 
 		 SEL_DAC				: out std_logic;
 		 SEL_EXT_A		  	: out std_logic;
 		 SEL_EXT_B			: out std_logic;
 		 SEL_MDAC2			: out std_logic;
		 
		 LED_1				: out std_logic;
		 LED_2				: out std_logic;
		 LED_3				: out std_logic;
		 LED_4				: out std_logic;
		 HOLD_1				: out std_logic;
		 HOLD_2				: out std_logic;
		 
		 Beam_Cyc_Reset 	: out std_logic
		);
end Cogging_Test_Switch;
	
architecture main of Cogging_Test_Switch is

signal test		: std_logic;
signal timer			: std_logic_vector(23 downto 0);
signal timer_ena		: std_logic;
signal test_mode		: std_logic;
signal cycle_reset	: std_logic;

	-- State Machine declarations ------------------------------------
	type STATE_TYPE is 
	(idle, switched, send_reset);
	signal CS, NS  : STATE_TYPE;

-- ## MAIN ##############################################################
begin 
EDGE_ON_TEST_SELECT_TRIGGER: led
port map(sig_in => Enable_Test, clk => clk, clrn => nreset, pls_out => test);


PROCESS_TIMER: process(clk, nreset, timer_ena, timer)
begin
	if  (nreset = '0') then 		timer <= (others => '0');
	elsif rising_edge(clk) then
		if  (timer_ena = '0') then 	timer <= (others => '0');
		else					timer <= timer + 1;
		end if;
	end if;
end process;


----------------------------------------------------------	
sm_reg: process (clk, nreset)	
begin
	if(nreset='0')              then CS <= idle;
	elsif(clk'event and clk='1')then CS <= NS;
	end if;
end process; -- sm_reg

sm_combinatorial: 
process (CS, timer, test, DIN_1, DIN_2, DIN_3, DIN_4,	T1, T2, T3, T4)
begin
	-- default outputs --
	timer_ena 		<= '1'; -- not enabled only when in idle state
	cycle_reset		<= '0';
	
	DO1			<= T1;  -- Normal Digital Inputs to Outputs
	DO2			<= T2;
	DO3			<= T3;
	DO4			<= T4;
		
	case CS is
		when idle   =>
			timer_ena		<= '0';  
			if (test = '1') then 		NS <= switched;	
			else					NS <= idle;
			end if;

		when switched =>
			if   (timer = X"30D400") then	NS <= send_reset; -- time out at 40 ms
			else					NS <= switched;
			end if;
			
			DO1			<= DIN_1; 
			DO2			<= DIN_2;
			DO3			<= DIN_3;
			DO4			<= DIN_4;
		
		when send_reset  =>			NS <= idle;
			cycle_reset	<= '1';
			
		when others =>				NS <= idle;
	end case; 
end process;

----------------------------------------------------------
SWITCH_CONTROL: process(clk, nreset, test, cycle_reset)
begin                              -- THIS ACTS AS AN SR-FLIP FLOP
	if    (nreset = '0')          then 	test_mode <= '0';
	elsif rising_edge(clk) then 
		if   (test = '1') 	then  test_mode <= '1';
		elsif(cycle_reset = '1')then  test_mode <= '0';
		end if;
	end if;
end process; 

	Beam_Cyc_Reset	<= cycle_reset;
	
 	SEL_DAC		<= '1';		
 	SEL_EXT_A		<= '0';	-- not used	
 	SEL_EXT_B		<= test_mode; -- a.k.a. SEL_MDAC1		
 	SEL_MDAC2		<= not test_mode;		

	LED_1			<= SW_1;		
	LED_2			<= LED_PB_SW;		
	LED_3			<= not test_mode;		
	LED_4			<= SW_2;	
	
	HOLD_1		<= '0'; -- not used
	HOLD_2		<= '0';
end main;

